Integer Partitions and Generating Functions

September 25, 2001

1 Introduction

For any positive integer n, a *partition* of n is a non-increasing sequence of positive integers $p_1, p_2, \ldots, p_n$ that add up to n. Each p_i is called a *part* of the partition. We will use $p(n)$ to denote the number of partitions of n. The following example shows all partitions of 6. These are generated using the *Combinatorica* function `Partitions`.

```
In[1]:= Partitions[6] // ColumnForm

Out[1]= {6}
        {5, 1}
        {4, 2}
        {4, 1, 1}
        {3, 3}
        {3, 2, 1}
        {3, 1, 1, 1}
        {2, 2, 2}
        {2, 2, 1, 1}
        {2, 1, 1, 1, 1}
        {1, 1, 1, 1, 1, 1}
```

Combinatorica also provides a function called `NumberOfPartitions` that computes $p(n)$. In the following example, $p(n)$ is shown for $n = 1, 2, \ldots, 30$. How rapidly does $p(n)$ grow? What is its asymptotic behavior? The answer to these questions is one the most celebrated outcomes of the collaboration between Hardy and Ramanujan.

```
In[2]:= Table[NumberOfPartitions[i], {i, 30}]

Out[2]= {1, 2, 3, 5, 7, 11, 15, 22, 30, 42, 56, 77, 101, 135, 176, 231, 297,

>    385, 490, 627, 792, 1002, 1255, 1575, 1958, 2436, 3010, 3718, 4565, 5604}
```

2 Generating Integer Partitions

Let $p(n,k)$ denote the number of partitions of n whose parts are no larger than k. For certain values of n and k, $p(n,k)$ is easy to compute. For example, $p(n,1) = p(1,k) = 1$ for all positive integers n and k. Also, $p(n,2) = \lfloor n/2 \rfloor + 1$ because the number of parts that are 2 is any number in $\{0, 1, 2, \ldots, \lfloor n/2 \rfloor\}$. Also, $p(n,k) = p(n,n)$ for all $k \geq n$. The set of partitions of n whose largest part is no greater than k can be partitioned into two sets: a set A containing partitions of n with largest part k and a set B containing partitions of n with largest part at most $(k-1)$. The size of B is $p(n, k-1)$. Removing k from each partition in A gives us the set of partitions of $n-k$ whose largest element is no greater than k. Thus the size of A is $p(n-k,k)$. This leads to the recurrence

$$p(n,k) = p(n,k-1) + p(n-k,k)$$

for all positive integers n and k, with $k \leq n$. We can use bases cases $p(n,0) = 0$ for all integers $n \geq 1$ and $p(0,k) = 1$ for all integers $k \geq 0$. Since $p(n) = p(n,n)$, this recurrence provides a way of counting the number of partitions of n and also provides an algorithm for generating partitions. Here is *Combinatorica* code for `Partitions`.

```
Partitions[n_Integer] := Partitions[n,n]
Partitions[n_Integer,_] := {} /; (n<0)
Partitions[0,_] := { {} }
Partitions[n_Integer,1] := { Table[1,{n}] }
Partitions[_,0] := {}

Partitions[n_Integer,maxpart_Integer] :=
        Join[
                Map[(Prepend[#,maxpart])&, Partitions[n-maxpart,maxpart]],
                Partitions[n,maxpart-1]
        ]
```

3 Generating Functions for Integer Partitions

A *generating function* is a function $G(z)$ whose form is

$$G(z) = g_0 + g_1 z + g_2 z^2 + \cdots = \sum_{n=0}^{\infty} g_n z^n.$$

$G(z)$ or in short G is said to be a generating function for the sequence $(g_0, g_1, g_2, \ldots)$. Generating functions provide the most powerful tool for dealing with sequences of numbers. For example, we can define a generating function $P(z) = \sum_{n=0}^{\infty} p(n) z^n$ for $(p(0), p(1), p(2), \ldots)$ and use this to derive many identities for integer partitions.

Let us start with a simpler example. Start with the function $(1+z)^n$ and rewrite it as

$$(1+z)^n = (1+z) \cdot (1+z) \cdots (1+z).$$

For the moment, assume that you don't know the Binomial Theorem and suppose that $(1+z)^n = B_n(z) = \sum_{i=0}^{\infty} b_{n,i} z^i$. Each term $b_{n,i} z^i$ is obtained from $(1+z) \cdot (1+z) \cdots (1+z)$ by picking z from i brackets and picking 1 from the remaining $(n-i)$ brackets. First, this means that $b_{n,i} = 0$ for any $i > n$. This also means that the coefficient $b_{n,i}$ of z^i is the number of ways of choosing i brackets from among n brackets to pick z from. This, as we know very well, is $\binom{n}{i}$. Thus

$$B_n(z) = \sum_{i=0}^{\infty} \binom{n}{i} z^i = (1+z)^n.$$

Another way of saying this is that $(1+z)^n$ is the generating function of the binomial sequence $(\binom{n}{0}, \binom{n}{1}, \binom{n}{2}, \ldots)$.

Let us now move on to the problem we are really interested in: finding a generating function for the sequence $(p(0), p(1), p(2), \ldots)$. We will now show that

$$(1 + z + z^2 + \cdots)(1 + z^2 + z^4 + \cdots)(1 + z^3 + z^6 + \cdots) \cdots \tag{1}$$

is a generating function for $(p(0), p(1), p(2), \ldots)$. Note that the i th bracket in the above product has the form

$$(1 + z^i + z^{2i} + z^{3i} + \cdots) = \sum_{c=0}^{\infty} z^{ci}.$$

As an example, consider how z^4 can be obtained from the above product. We see that z^4 can be obtained by

(i) picking z^4 from bracket 4 and 1 from every other bracket,

(ii) picking z^3 from bracket 3, z from bracket 1, and 1 from every other bracket,

(iii) picking z^4 from bracket 2, and 1 from every other bracket,

(iv) picking z^2 from bracket 2, z^2 from bracket 1, and 1 from every other bracket,

(v) picking z^4 from bracket 1, and 1 from every other bracket.

So z^4 can be obtained in 5 ways and therefore the coefficient of z^4 in the expansion of (1) is 5. In general, the number of ways of obtaining z^n is equal to the number of ways of choosing $z^{1 \cdot c_1}, z^{2 \cdot c_2}, z^{3 \cdot c_3}, \ldots$ respectively from brackets $1, 2, 3, \ldots$ so that $1 \cdot c_1 + 2 \cdot c_2 + 3 \cdot c_3 + \cdots = n$. Interpreting c_i as the number of copies of i chosen, we see this as the number of ways of partitioning n.

Thus

$$(1 + z + z^2 + \cdots)(1 + z^2 + z^4 + \cdots)(1 + z^3 + z^6 + \cdots)\cdots$$

is the generating function $P(z) = \sum_{n=0}^{\infty} p(n) z^n$. Rewriting the geometric series $(1 + z^i + z^{2i} + z^{3i} + \cdots)$ as $1/(1 - z^i)$ we get that

$$P(z) = \prod_{i=1}^{\infty} \frac{1}{(1 - z^i)}. \tag{2}$$

This beautiful result is due to Euler.

Mathematica provides considerable amount of machinery for dealing with generating functions. To start with, here is a example of a function called `Series` that returns the power series expansion of a function.

```
In[3]:= Series[1/(1-z^3), {z, 0, 30}]

                3      6      9      12      15      18      21      24      27      30       31
Out[3]= 1 + z  + z  + z  + z  + z  + z  + z  + z  + z  + z  + O[z]
```

In the following experiment, we use `Series` and the *Mathematica* function `Product` to verify that the first 20 terms in the expansion of $\prod_{i=1}^{\infty} 1/(1 - z^i)$ have the correct coefficients (check this!).

```
In[4]:= Product[Series[1/(1 - x^i), {x, 0, 20}], {i, 20}]

                       2       3       4       5        6        7        8        9
Out[4]= 1 + x + 2 x  + 3 x  + 5 x  + 7 x  + 11 x  + 15 x  + 22 x  + 30 x  +

            10         11         12          13          14          15          16
>     42 x   + 56 x   + 77 x   + 101 x   + 135 x   + 176 x   + 231 x   +

             17          18          19          20         21
>     297 x   + 385 x   + 490 x   + 627 x   + O[x]
```

So we have a pretty generating function for the number of partitions, how is this useful to us? In the following examples, I will show you how we can use this generating function to derive several identities.

Example 1. What is the number of partitions of n that do not contain 1? Let $q(n)$ denote this quantity and let $Q(z) = \sum_{n=0}^{\infty} q(n)z^n$ denote the generating function for $(q(0), q(1), q(2), \ldots)$. From the discussion that preceded (2) we see that $Q(z)$ is identical to $P(z)$ except that $Q(z)$ does not have the first term in the product, that is, the term that contributes copies of 1. So $Q(z) = \prod_{i=2}^{\infty} 1/(1-z^i) = (1-z)P(z)$. Now,

$$\begin{aligned}
(1-z)P(z) &= \sum_{n=0}^{\infty} p(n) \cdot z^n - \sum_{n=0}^{\infty} p(n) \cdot z^{n+1} \\
&= \sum_{n=0}^{\infty} p(n) \cdot z^n - \sum_{n=0}^{\infty} p(n-1) \cdot z^n \\
&= \sum_{n=0}^{\infty} (p(n) - p(n-1))z^n.
\end{aligned}$$

Note that in the second equation above $p(-1)$ occurs and this is assumed to be 0. This implies that $q(n) = p(n) - p(n-1)$. In other words, the number of partitions of n which do not contain 1 equals the number of partitions of n minus the number of partitions of $(n-1)$. Can you come up with a bijective proof of this?

Example 2. What is the number of partitions of n in which the parts are at most k? We have used $p(n,k)$ to denote this quantity and now we let $R_k(z) = \sum_{n=0}^{\infty} p(n,k)z^n$ be the generating function for $(p(0,k), p(1,k), p(2,k), \ldots)$. From the discussion preceding (2) we know that $R_k(z) = \prod_{i=1}^{k} 1/(1-z^i)$ because in partitions we are considering there are no parts larger than k. This implies that $R_{k-1}(z) = (1-z^k)R_k(z)$. Equating coefficients of like terms on both sides of the equation, we get $p(n,k-1) = p(n,k) - p(n-k,k)$. This is identical to the recurrence $p(n,k) = p(n,k-1) + p(n-k,k)$ that we gave bijective proof for earlier.

Example 3. In this example, we will use the generating function $P(z)$ to derive the remarkable identity

$$p(n) = \frac{1}{n} \sum_{k=1}^{n} \sigma(k)p(n-k).$$

Here $\sigma(k)$ is the sum of the factors of k. Start with $P(z) = \prod_{i=1}^{\infty} 1/(1-z^i)$ and take logarithms on both sides to get

$$\log(P(z)) = \sum_{i=1}^{\infty} \log\left(\frac{1}{(1-z^i)}\right).$$

Then differentiate both sides with respect to z and move $P(z)$ to the right to get

$$P'(z) = P(z) \cdot \sum_{i=1}^{\infty} \frac{i \cdot z^{i-1}}{(1-z^i)}. \tag{3}$$

Now let us focus on $\sum_{i=1}^{\infty} (i \cdot z^{i-1})/(1-z^i)$ and simplify it as follows:

$$\sum_{i=1}^{\infty} \frac{i \cdot z^{i-1}}{(1-z^i)} = \sum_{i=1}^{\infty} (i \cdot z^{i-1}) \cdot (1 + z^i + z^{2i} + z^{3i} + \cdots) = \sum_{i=1}^{\infty} i \cdot z^{i-1} \cdot \sum_{j=0}^{\infty} z^{ji}.$$

Further simplification leads to

$$\sum_{i=1}^{\infty} i \cdot z^{i-1} \cdot \sum_{j=1}^{\infty} z^{(j-1)i} = \sum_{i=1}^{\infty} \sum_{j=1}^{\infty} i \cdot z^{(ij-1)}.$$

In this sum, which terms contribute to z^k? Any term z^{ij-1} with $ij-1=k$ contributes i to the coefficient of z^k. In other words, any i that is a factor of $k+1$ contributes to coefficient of z^k. Thus the coefficient of z^k is $\sigma(k+1)$. The above sum can therefore be rewritten as

$$\sum_{k=0}^{\infty} \sigma(k+1) z^k.$$

So the equation (3) can be rewritten as

$$P'(z) = P(z) \cdot \sum_{k=0}^{\infty} \sigma(k+1) z^k.$$

Now we need to equate coefficients of like terms on the two sides of the equation. The left hand side is

$$P'(z) = p(1) + 2zp(2) + 3z^2 p(3) + \cdots = \sum_{n=1}^{\infty} (n+1) p(n+1) z^n.$$

The right hand side is

$$P(z) \cdot \sum_{k=1}^{\infty} \sigma(k+1) z^k = \sum_{\ell=0}^{\infty} p(\ell) z^\ell \cdot \sum_{k=0}^{\infty} \sigma(k+1) z^k = \sum_{n=0}^{\infty} \left[\sum_{k=0}^{n} \sigma(k+1) p(n-k) \right] z^n.$$

Equating the coefficients of z^n we get

$$(n+1)p(n+1) = \sum_{k=0}^{n} \sigma(k+1) p(n-k).$$

This is equivalent to

$$np(n) = \sum_{k=0}^{n-1} \sigma(k+1) p(n-(k+1))$$

and this is equivalent to

$$p(n) = \frac{1}{n} \sum_{k=1}^{n} \sigma(k) p(n-k).$$

Example 4. This example shows that the number of partitions with all odd parts is identical to the number of partitions with distinct parts. In the following example, we select all partitions of 11 with distinct parts. We see that there are 12 of these.

```
In[5]:= Select[Partitions[11], Length[Union[#]] == Length[#] &]

Out[5]= {{11}, {10, 1}, {9, 2}, {8, 3}, {8, 2, 1}, {7, 4}, {7, 3, 1}, {6, 5},

>    {6, 4, 1}, {6, 3, 2}, {5, 4, 2}, {5, 3, 2, 1}}
```

In the following example, we select all partitions of 11 with whose parts are all odd. Again, we see that there are 12 of these.

```
In[6]:= Select[Partitions[11], Apply[And, Map[Function[x, OddQ[x]], #]] &]

Out[6]= {{11}, {9, 1, 1}, {7, 3, 1}, {7, 1, 1, 1, 1}, {5, 5, 1}, {5, 3, 3},

>    {5, 3, 1, 1, 1}, {5, 1, 1, 1, 1, 1, 1}, {3, 3, 3, 1, 1},
```

```
>    {3, 3, 1, 1, 1, 1, 1}, {3, 1, 1, 1, 1, 1, 1, 1, 1},

>    {1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1}}
```

This is not a coincidence as the following proof shows.

The generating function for the number of partitions with distinct parts is $(1+x)(1+x^2)(1+x^3)\cdots$. We rewrite it as

$$\frac{(1-x^2)}{(1-x)}\frac{(1-x^4)}{(1-x^2)}\frac{(1-x^6)}{(1-x^3)}\cdots = \frac{1}{(1-x)}\frac{1}{(1-x^3)}\frac{1}{(1-x^5)}\cdots$$

The right hand side of the above equation is the generating function for partitions all of whose parts are odd. This is a slick proof using generating functions. Can you devise a bijective proof?

4 Euler's Pentagonal Theorem

About 80 years ago Percy MacMahon computed the values of $p(n)$ for $n = 1, 2, 3, \ldots, 200$, by hand. This turned out to be immensely useful for Hardy and Ramanujan who were trying to check how accurate their formula for approximating $p(n)$ was. In particular, MacMahon found that $p(200) = 3,972,999,029,388$. This is verified by *Combinatorica* below.

```
In[7]:= NumberOfPartitions[200]

Out[7]= 3972999029388
```

MacMahon could have used the recurrence $p(n,k) = p(n-k,k) + p(n,k-1)$ to calculate $p(n,n)$ or he could have used the the recurrence $p(n) = 1/n\sum_{k=1}^{n}\sigma(k)p(n-k)$. However, it turns out that there is a recurrence that provides a much faster way of computing $p(n)$. This is given by *Euler's Pentagonal Theorem.*

Consider the function

$$\prod_{i=1}^{\infty}\frac{1}{(1-z^i)}.$$

What is the combinatorial significance of this function? In other words, if $\prod_{i=1}^{\infty}\frac{1}{(1-z^i)} = \sum_{n=0}^{\infty}q(n)z^n$, is there some relevant combinatorial interpretation of the sequence $(q(0), q(1), q(2), \ldots)$? To answer this consider the more familiar expression

$$\prod_{i=1}^{\infty}\frac{1}{(1+z^i)}.$$

As we know by now, this is the generating function for the sequence $(pd(0), pd(1), pd(2), \ldots)$, where $pd(n)$ is the number of ways of partitioning n into distinct parts. Thus $pd(n)$, the coefficent of z^n, can be written as $1+1+1+\cdots+1$ where each 1 corresponds to a way of picking $z^{i_1}, z^{i_2}, z^{i_3}, \ldots, z^{i_k}$ such that $n = i_1 + i_2 + \cdots + i_k$. Similarly, $q(n)$, the coefficient of z^n in $\prod_{i=1}^{\infty}1/(1-z^i)$, can be written as $(1+1+\cdots+1)+(-1-1-\cdots-1)$, where the $+1$'s correspond to ways of picking $z^{i_1}, z^{i_2}, z^{i_3}, \ldots, z^{i_k}$ such that $n = i_1 + i_2 + \cdots + i_k$, for even k and the -1's correspond to the ways of picking $z^{i_1}, z^{i_2}, z^{i_3}, \ldots, z^{i_k}$ such that $n = i_1 + i_2 + \cdots + i_k$ for odd k. Thus the coefficient $q(n) = pde(n) - pdo(n)$ where $pde(n)$ is the number of ways of partitioning n into an even number of distinct parts and $pdo(n)$ is the number of ways of partitioning n into an odd number of distinct parts.

Euler showed that

$$\prod_{i=1}^{\infty} \frac{1}{(1-z^i)} = 1 - z - z^2 + z^5 + z^7 - z^{12} - z^{15} + \cdots .$$

The sequence of exponents $0, 1, 2, 5, 7, 12, 15, \ldots$ that corresponds to non-zero terms in the right hand side above is the well-known sequence of *pentagonal numbers*. The pentagonal numbers have the form $k(3k+1)/2$ for integers k and the coefficient of the terms $z^{k(3k+1)/2}$ is $+1$ if k is even and is -1 otherwise. Thus Euler's expansion formula can be rewritten as

$$\prod_{i=1}^{\infty} \frac{1}{(1-z^i)} = \sum_{k=-\infty}^{\infty} (-1)^k z^{k(3k+1)/2}.$$

In the example below I calculate the above product partially, multiplying the first 10 terms. The resulting polynomial is correct in the coefficients of z^n for any n, $0 \le n \le 10$ since these terms will not be affected by any subsequent multiplication. Note that some of the remaining terms have coefficients that are not in the set $\{-1, 0, 1\}$.

```
In[8]:=  Expand[Product[(1 - z^i), {i, 10}]]

                       2     5     7      11      12      13      14        15      18      19
Out[8]= 1 - z - z  + z  + z  + z   - z   - z   - z   - 2 z   + z   + z   +

      20      21        22      25      26        27        28      29      30        33
>    z   + z   + 3 z   - z   - z   - 2 z   - 2 z   - z   - z   + 3 z   +

      34      35      36      37        40      41      42      43      44      48      50
>    z   + z   + z   + z   - 2 z   - z   - z   - z   + z   + z   + z   -

      53      54      55
>    z   - z   + z
```

The combinatorial implication of what Euler showed is that

$$pde(n) - pdo(n) = \begin{cases} 0 & \text{if } n \neq k(3k+1)/2 \text{ for any integer } k \\ (-1)^k & \text{if } n = k(3k+1)/2 \text{ for some integer } k \end{cases}$$

A more important combinatorial implication of Euler's Theorem is derived as follows. First, for convenience let $t(n)$ denote $pde(n) - pdo(n)$.

$$\begin{aligned} 1 &= \prod_{i=1}^{\infty} \frac{1}{(1-x^i)} \cdot \prod_{i=1}^{\infty} (1-x^i) \\ &= \left(\sum_{n=0}^{\infty} p(n) z^n \right) \cdot \left(\sum_{n=0}^{\infty} t(n) z^n \right) \end{aligned}$$

For any integer $k > 0$, the coefficient of k on the left hand side is 0 while the coefficient on the right hand side is

$$\sum_{i=0}^{n} p(i)t(i) = p(n) - p(n-1) - p(n-2) + p(n-5) + p(n-7) + \cdots$$

Equating the two coefficients we get

$$p(n) = p(n-1) + p(n-2) - p(n-5) - p(n-7) + \cdots$$

In other words,

$$p(n) = \sum_{k=1}^{\infty} (-1)^{k+1} \Big(p(n - k(3k+1)/2) + p(n + k(3k+1)/2) \Big).$$

This is an infinite series but only has $O(\sqrt{n})$ non-zero terms. Here is the *Combinatorica* implementation of the function `NumberOfPermutations`. This uses the recurrence just derived.

```
NumberOfPartitions1[n_Integer] := 0  /; (n < 0)
NumberOfPartitions1[n_Integer] := 1  /; (n == 0)

NumberOfPartitions1[n_Integer] := NumberOfPartitions1[n] =
        Module[{m},
                Sum[ (-1)^(m+1) NumberOfPartitions1[n - m (3m-1)/2] +
                        (-1)^(m+1) NumberOfPartitions1[n - m (3m+1)/2],
                        {m, Ceiling[ (1+Sqrt[1.0 + 24n])/6 ], 1, -1}
                ]
        ]
```

Using this implementation it is easy to do better than Percy MacMahon.

```
In[9]:= NumberOfPartitions[700]

Out[9]= 60378285202834474611028659
```